

教学参考〈五〉

# BASIC

## 程序例题选

谭浩强 周朝龙

上海电大通讯编辑室

## 前 言

BASIC 是得到广泛使用的计算机语言。现在随着计算机在国内逐步推广普及,学习 BASIC 的人愈来愈多了。在初步学习 BASIC 语言以后,应当逐步提高自己的编制程序的技巧。

本例题选提供了一百个程序例子,力求将实用性、通用性、趣味性于一炉。对一些较难的题目,附以数学分析、推导过程和必要的说明,以帮助初步学过 BASIC 的读者提高编制程序的技巧,作为学习 BASIC 的参考材料。书中题目可以作为例题讲授,也可以选择一部分作为习题。读者可以通过对各种例子的学习和比较,了解如何使用 BASIC 解决实际问题。当然,读者还可能编写出更好的程序。

本书由清华大学谭浩强同志修改、审定。西北轻工业学院周朝龙同志在编写过程中,曾得到该校焦尚仁等同志的帮助。限于时间和作者的水平,错误或不当之处,恳请读者批评指正。

编 者

一九八二年五月

# 目 录

## 前言

## 第一部分

基本 BASIC 语句的使用 ..... 1—58

## 第二部分

常用的计算方法 ..... 59—117

## 第三部分

使用 DJS-130 扩展 BASIC 语句编写的程序 ... 118—150

## 第四部分

实际应用(包括计算机游戏等)举例 ..... 181—188

## 第五部分

控制绘图机画图程序 ..... 189—202

## 第一部分

【一】 求  $y = 2^0 + 2^1 + 2^2 + \dots + 2^{60}$

解 1:

```
10 LET Y = 1
20 FOR I = 1 TO 60
30 LET Y = Y + 2 ↑ I
40 NEXT I
50 PRINT "Y = "; Y
60 END
```

解 2:

考察和式各项, 后一项应为前一项乘 2。即,

$$a_n = a_{n-1} \cdot 2$$

倘计算  $a_{n-1}$  项后能将  $a_{n-1}$  之值保存好, 则在求  $a_n$  项时, 就能大大节省计算时间, 故有如下程序:

```
5 LET A = 1 (令  $A = a_0 = 2^0 = 1$ )
10 LET Y = A (令  $Y = a_0 = A$ )
20 FOR N = 1 TO 60
30 LET A = A * 2 (让  $a_n = a_{n-1} \cdot 2$ )
40 LET Y = Y + A ( $y = \sum_{n=0}^{60} a_n$ )
50 NEXT N
60 PRINT "Y = ", Y
70 END
```

[二] 求  $y = 1 - 2^2 + 3^2 - 4^2 + \dots + 99^2 - 100^2$

解1: 原式可化为

$$y = (1^2 + 3^2 + 5^2 + \dots + 99^2) - (2^2 + 4^2 + 6^2 + \dots + 100^2)$$

令  $A = 1^2 + 3^2 + \dots + 99^2$ ,  $B = 2^2 + 4^2 + \dots + 100^2$ ,

则  $Y = A - B$

程序如下:

```
10 LET A=0
20 FOR N=1 TO 99 STEP 2
30 LET A=A+N^2
40 NEXT N
50 LET B=0
60 FOR N=2 TO 100 STEP 2
70 LET B=B+N^2
80 NEXT N
90 LET Y=A-B
100 PRINT "Y="; Y
110 END
```

解2: 观察原式右边, 可建立如下通项公式:

$$a_n = (-1)^{n+1} \cdot n^2 \quad (\text{其中 } n=1, 2, \dots, 100)$$

故程序可简化为:

```
10 LET Y=0
20 FOR N=1 TO 100
30 LET Y=Y+(-1)^(N+1)*N^2
40 NEXT N
50 PRINT "Y="; Y
60 END
```

3: 原式还可化为

$$y = (1^2 - 2^2) + (3^2 - 4^2) + \dots + (99^2 - 100^2)$$

其通项式为：

$$a_n = n^2 - (n+1)^2 = -(2n+1) \quad (\text{其中 } n=1, 3, 5, \dots, 99)$$

故程序还可这样编写：

```

10 LET Y=0
20 FOR N=1 TO 99 STEP 2
30 LET Y=Y-(2*N+1)
40 NEXT N
50 PRINT "Y="; Y
60 END

```

[三] 求  $y = (1+2+3+4) + (6+7+8+9) + \dots$   
 $+ (96+97+98+99)$

解1：原式可化为

$$\begin{aligned}
 y &= (1+2+3+\dots+99) - (5+10+15+\dots+95) \\
 &= \sum a_i - \sum b_i
 \end{aligned}$$

程序如下：

```

10 LET A=0
20 FOR N=1 TO 99
30 LET A=A+N
40 NEXT A
50 LET B=0
60 FOR N=5 TO 95 STEP 5
70 LET B=B+N
80 NEXT N
90 LET Y=A-B (y=ΣA(I)-ΣB(I))
100 PRINT "Y="; Y
110 END

```

解2: 原式可化为

$$y = (0 + 10) + (20 + 10) + (40 + 10) + \dots + (380 + 10)$$

程序如下:

```
10 LET Y = 0
20 FOR N = 0 TO 380 STEP 20
30 LET Y = Y + (N + 10)
40 NEXT N
50 PRINT "Y = ", Y
60 END
```

解3: 原式还可化简为

$$y = 10 + 30 + 50 + \dots + 390$$

程序如下:

```
10 LET Y = 0
20 FOR N = 10 TO 390 STEP 20
30 LET Y = Y + N
40 NEXT N
50 PRINT "Y = ", Y
60 END
```

解4: 原式可化为

$$y = A + B$$

其中:

$$\begin{aligned} A &= \sum \left\{ \begin{array}{l} 1 + 11 + 21 + \dots + 91 \\ 2 + 12 + 22 + \dots + 92 \\ 3 + 13 + 23 + \dots + 93 \\ 4 + 14 + 24 + \dots + 94 \end{array} \right\} \\ &= \sum_{i=1}^4 a_{ij} \quad \left( \begin{array}{l} i = 1, 2, 3, 4 \\ j = i, i + 10, \dots, i + 90 \end{array} \right) \end{aligned}$$

$$B = 2 \left\{ \begin{array}{l} 6 + 16 + 26 + \dots + 96 \\ 7 + 17 + 27 + \dots + 97 \\ 8 + 18 + 28 + \dots + 98 \\ 9 + 19 + 29 + \dots + 99 \end{array} \right\}$$

$$= \sum_{i=6}^9 \sum_{j=i, i+10, \dots, i+90}$$

程序如下：

```

10 LET A=0
20 FOR I=1 TO 4
30 FOR J=I TO I+90 STEP 10
40 LET A=A+J
50 NEXT J
60 NEXT I
70 LET B=0
80 FOR I=6 TO 9
90 FOR J=I TO I+90 STEP 10
100 LET B=B+J
110 NEXT J
120 NEXT I
130 PRINT "Y=", A+B
140 END

```

求 A

求 B

解 5：原式可化为

$$y = 1 + 2 + 3 + \dots + 99$$

当某项为 5 的倍数时则跳过不加。即：若某项  $I = INT(I/5)*5$  时，则  $I$  必为 5 的倍数，可用条件语句来实现。故程序又可编写为：

```

10 LET Y=0

```

```

20 FOR I=1 TO 99
30 IF INT(I/5)*5=I THEN 50
40 LET Y=Y+I
50 NEXT I
60 PRINT "Y="; Y
70 END

```

解6: 原式右边的通项式可改写为

$$a_n = 5I + J \quad (n = 1, 2, \dots, 20)$$

其中:  $I = 0, 1, 2, \dots, 19$

对应每个  $I$  值,  $J = 1, 2, 3, 4$

所以程序还可这样编写:

```

10 LET Y=0
20 FOR I=0 TO 19
30 FOR J=1 TO 4
40 LET Y=Y+(5*I+J)
50 NEXT J
60 NEXT I
70 PRINT "Y="; Y
80 END

```

由上可知: 对某些题目来说, 程序的编写可以有多种多样, 总的原则是:

- (1) 使程序简单, 清晰, 易于理解。
- (2) 使计算速度最快(即占用主机时间最短);

[四] 有  $N$  个非零数据, 试求

- (1) 负数之和及负数个数
- (2) 正数之和及正数个数

解: 变量单元设置如下:

A——存放负数之和

B——存放正数之和

M——存放负数个数(正数个数应为  $N-M$ )

程序如下:

```
5  PRINT "N=";
10  INPUT N  (键盘输入 N 之值, N——数据个数)
15  PRINT
20  LET A=0
30  LET B=0
40  LET M=0
50  FOR I=1 TO N
60  READ X
70  IF X>0 GOTO 110
80  LET A=A+X  (负数累加)
90  LET M=M+1  (负数个数计数器)
100 GOTO 120
110 LET B=B+X  (正数累加)
120 NEXT I
130 PRINT "A="; A, "NUMBER:"; M
140 PRINT "B="; B, "NUMBER:"; N-M
150 END
200 DATA } 填写上 n 个数据
    :
```

[五] 计算并打印函数

$$y = 3.14x^2 + 5.4x + 7.6$$

在区间  $[0.01, 0.02]$  内 11 个等分点处的函数值。要求打印格式分别为:

|     |            |             |
|-----|------------|-------------|
| (1) | X          | Y           |
|     | 0.01       | .....       |
|     | 0.011      | .....       |
|     | ⋮          | ⋮           |
|     | 0.02       | .....       |
| (2) | X(1)=0.01  | Y(1)=.....  |
|     | X(2)=0.011 | Y(2)=.....  |
|     | ⋮          | ⋮           |
|     | X(11)=0.02 | Y(11)=..... |

解 1: (按第一种打印格式要求编写程序)

```

10 PRINT "X", "Y"
20 PRINT
30 FOR X=0.01 TO 0.02 STEP 0.001
40 LET Y=3.14*X*X+5.4*X+7.5
50 PRINT X, Y
60 NEXT X
70 END

```

解 2: (按第二种打印格式要求编写程序)

```

5 LET I=0 (计数器置零)
10 FOR X=0.01 TO 0.02 STEP 0.001
15 LET I=I+1 (设置计数器)
20 LET Y=3.14*X*X+5.4*X+7.5
30 PRINT "X(", I, ")=", X, "Y(", I, ")=", Y
40 NEXT X
50 END

```

[六] 求  $y = \frac{x^2}{1!} + \frac{x^3}{2!} + \frac{x^4}{3!} + \dots + \frac{x^{n+1}}{n!}$  ( $x, n$  为已知)

解 1: 将求阶乘编为子程序, 然后不断调用子程序即可求得  $y$  值。

程序如下:

```
10 PRINT "X, N";  
20 INPUT X, N  
30 PRINT  
40 LET Y = 0  
50 FOR A = 1 TO N  
60 GOSUB 200 (转去计算阶乘 a!)  
70 LET Y = Y + X ↑ (A + 1) / S  
80 NEXT A  
90 PRINT "Y = ", Y  
100 END  
200 REM SUBROUTINE  
210 LET S = 1  
220 FOR B = 1 TO A  
230 LET S = S * B  
240 NEXT B  
250 RETURN
```

其中 200~250 语句即为求阶乘子程序。

本程序的优点在于结构清楚, 层次分明。但其明显缺点是:

- (1) 每次求阶乘均需从 1 开始乘起;
- (2) 当  $x < 0$  时, BASIC 解释程序将视  $X \uparrow (A + 1)$  为无定义。

故有解 2。

解 2,  $\because n! = (n - 1)! \cdot n$

$$x^n = x^{n-1} \cdot x$$

∴ 若在计算  $(n-1)!$  和  $x^{n-1}$  时分别将其结果保存好，则称为计算  $n!$  及  $x^n$  之值时提供极大方便，从而大大提高计算速度。

改进后的程序如下：

```

10 PRINT "X, N";
20 INPUT X, N
30 PRINT
40 LET Y = 0
50 LET A = 1
60 LET W = X
70 FOR I = 1 TO N
80 LET A = A * I      (求  $n! = (n-1)! \cdot n$ )
90 LET W = W * X      (求  $x^n = x^{n-1} \cdot x$ )
100 LET Y = Y + W / A  (求  $y = \sum_{i=1}^n \frac{x^{i+1}}{i!}$ )
110 NEXT I
120 PRINT "Y = ", Y
130 END

```

解 3. 若将原式化为如下形式：

$$\begin{aligned}
 y = & (x) \cdot \frac{x}{1} + \left(\frac{x^2}{1!}\right)\left(\frac{x}{2}\right) + \left(\frac{x^3}{2!}\right)\left(\frac{x}{3}\right) + \dots \\
 & + \left(\frac{x^n}{(n-1)!}\right)\left(\frac{x}{n}\right)
 \end{aligned}$$

这样，通项式可写为：

$$a_n = a_{n-1} \cdot \frac{x}{n}$$

即：后项 = (前项)  $\cdot \frac{x}{n}$

程序如下：

```

10  PRINT "X, N";
20  INPUT X, N
30  PRINT
40  LET Y=0
50  LET W=X
60  FOR I=1 TO N
70  LET W=W* X/I  (求  $a_n = a_{n-1} \cdot \frac{x}{n}$ )
80  LET Y=Y+W    (求  $y = \sum a_n$ )
90  NEXT I
100 PRINT "Y="; Y
110 END

```

$$\begin{aligned}
 \text{[七]} \quad \sin x = & \frac{x}{1} - \frac{x^3}{3!} + \frac{x^5}{5!} - \frac{x^7}{7!} + \dots \\
 & - \frac{x^{4n-1}}{(4n-1)!} + \frac{x^{4n+1}}{(4n+1)!}
 \end{aligned}$$

其中  $x, n$  为已知。

解 1：原式可化为：

$$\begin{aligned}
 \sin x = & \left[ \frac{x}{1} + \frac{x^5}{5!} + \frac{x^9}{9!} + \dots + \frac{x^{4n+1}}{(4n+1)!} \right] \\
 & - \left[ \frac{x^3}{3!} + \frac{x^7}{7!} + \dots + \frac{x^{4n-1}}{(4n-1)!} \right] = A - B
 \end{aligned}$$

程序如下：

```

10  PRINT "X, N";
20  INPUT X, N
30  PRINT
40  LET A=0

```

```

50 FOR I=1 TO 4*N+1 STEP 4
60 LET T=1
70 LET S=1
80 FOR J=1 TO I
90 LET T=T*I
100 LET S=S*X
110 NEXT J
120 LET A=.A+S/T
130 NEXT I
140 LET B=0
150 FOR I=3 TO 4*N-1 STEP 4
160 LET T=1
170 LET S=1
180 FOR J=1 TO I
190 LET T=T*J
200 LET S=S*X
210 NEXT J
220 LET B=B+S/T
230 NEXT I
240 PRINT "SIN(„X,„)=„, A-B
250 END

```

解2, 原式右边各项的通项式为:

$$a_i = (-1)^i \cdot \frac{x^{2i+1}}{(2i+1)!} \quad (i=0, 1, 2, \dots, n)$$

程序如下:

```

10 PRINT "X, N",
20 INPUT X, N

```

```

30 PRINT
40 LET Y=X (令 y 初值为  $a_0=x$ )
45 LET F=1
    (令  $F_0=1$ , 则  $F_i=(-1) \cdot F_{i-1}=(-1)^i$ )
50 FOR I=1 TO N
60 LET T=1
70 LET S=1
80 FOR J=1 TO 2*I+1
90 LET T=T*J
100 LET S=S*X
110 NEXT J
115 LET F=(-1)*F
120 LET Y=Y+F*S/T
130 NEXT I
140 PRINT "SIN(", X, ")="; Y
150 END

```

解3: 原式化为:

$$\sin x = x + \left[ -\frac{x^3}{3!} + \frac{x^5}{5!} - \dots + \frac{x^{4i+1}}{(4i+1)!} \right]$$

其中括号内各项的通项式为:

$$a_i = (-1)^i \cdot a_{i-1} \cdot \frac{x^2}{(i+1)(i+2)};$$

这里  $a_0=x$ ,  $i=1, 2, 3, \dots, n$

这样,在计算后一项时便可调用前一项的结果。程序如

下:

```

10 PRINT "X, N";
20 INPUT X, N

```

```

30 PRINT
40 LET Y=X (令 y 初值为原式第一项 x)
45 LET F=1
50 LET T=1
60 LET S=X (a0=x)
70 FOR I=1 TO N
80 LET T=T*(I+1)*(I+2)
90 LET S=S*X*X
95 LET F=(-1)*F
100 LET Y=Y+F*S/T
110 NEXT I
120 PRINT "SIN(", X, ") = ", Y
130 END

```

【八】从无穷乘积  $y = \frac{2}{1} \cdot \frac{2}{3} \cdot \frac{4}{3} \cdot \frac{4}{5} \cdot \frac{6}{5} \cdot \frac{6}{7} \cdots$  中取前 200

项计算 y 值。

解 1: 原式化为

$$y = \left[ \frac{2}{1} \cdot \frac{4}{3} \cdot \frac{6}{5} \cdots \left( \frac{2n}{2n-1} \right) \right] \cdot \left[ \frac{2}{3} \cdot \frac{4}{5} \cdot \frac{6}{7} \cdots \left( \frac{2n}{2n+1} \right) \right]$$

(其中  $n=1, 2, 3, \dots, 100$ )。

程序如下:

```

10 LET A=1
20 LET B=1
30 FOR N=1 TO 100
40 LET A=A*2*N/(2*N-1)
50 LET B=B*2*N/(2*N+1)
60 NEXT N

```